


11.3. Oscylator nieliniowy Duffinga

Dotychczas pokazane metody rozwiązywania dotyczyły liniowych równań różniczkowych. W niniejszym punkcie pokażemy przykład poszukiwania rozwiązania nieliniowego równania różniczkowego rzędu drugiego. Przedmiotem zainteresowania jest nieliniowe równanie Duffinga⁴, opisujące ruch nieliniowego oscylatora (rys. 11.6). W oryginalnej pracy Duffinga [13]



Rys. 11.6. Oryginalny układ rozważany w pracy G. Duffinga [13]

równanie różniczkowe opisujące ruch układu z rys. 11.6 ma postać:

$$\ddot{x} + \eta x - \theta x^2 - \chi x^3 = \kappa \sin(\omega t), \quad (11.17)$$

gdzie: $\eta = \frac{gS}{J} \cos(\Psi_0) + \frac{c^2}{K}(f_1 + f_2 + f_3)$,
 $\theta = \frac{1}{2L} \frac{gS}{J} \sin(\Psi_0)$,
 $\chi = \frac{1}{6L^2} \frac{gS}{J} \cos(\Psi_0)$,
 $\kappa = \frac{c f_1 L}{J} a$,
 J – moment bezwładności masy m ,
 gS – moment statyczny wahadła.

Zwróćmy uwagę, że w równaniu (11.17) nie pojawia się człon opisujący tłumienie, tak charakterystyczny dla uogólnionego równania Duffinga, spotykanego najczęściej w literaturze:

$$\ddot{x} + \dot{x}\varepsilon k + x + \varepsilon\alpha x^3 = \varepsilon F \cos(\Omega t). \quad (11.18)$$

⁴ Georg Duffing (1861–1944) – niemiecki inżynier i wynalazca. Zajmował się badaniem drgań i rezonansu. Sformułował nieliniowe równanie różniczkowe nazwane później jego nazwiskiem.

Równania (11.17) i (11.18) są przykładami chaotycznego oscylatora nieliniowego. Chaotyczność jego ruchu polega na tym, że niewielka zmiana parametrów układu powoduje znaczące zmiany w jego rozwiązaniu. Drugą cechą wyróżniającą to równanie jest jego złe uwarunkowanie, co oznacza, że użycie dwu różnych metod obliczeniowych prowadzi do odmiennych rezultatów. Rozwiązanie równań (11.17) i (11.18) bezpośrednio z wykorzystaniem poleceń `ode2()`, `cotrib_ode()` jest skazane na niepowodzenie.

11.3.1. Podejście numeryczne

W niniejszym punkcie skupimy się na rozwiązaniu numerycznym równania (11.17). Jeśli wprowadzimy oznaczenia:

$$\begin{aligned} x_1 &= x, \\ x_2 &= \frac{dx_1}{dt}, \end{aligned} \quad (11.19)$$


to równanie (11.17) możemy zapisać w postaci:

$$\frac{d}{dt} \begin{pmatrix} x_1 \\ x_2 \end{pmatrix} = \begin{pmatrix} x_2 \\ -\eta x_1 + \theta x_1^2 + \chi x_1^3 - \kappa \sin(\omega t) \end{pmatrix}. \quad (11.20)$$


Skrypt w Maximize rozwiązujący równanie (11.17) ma postać:

```
(%i1) kill(all)$ reset()$
(%i6) %theta:0.1$ %chi:-0.25$ %kappa:2.5$ %eta:-1$ %omega:2$;
(%i7) rownanie:[x2,-%eta*x1+%theta*x1^2+%chi*x1^3+
    %kappa*sin(%omega*t)];
(%i8) zmienne:[x1,x2]$
(%i11) wart_pocz:[%pi/2,0]$ iteracje:20$podzial:31$
(%i12) zakres_parametr:[t,0,iteracje*pi,%pi/podzial]$
(%i13) sol:rk(rownanie,zmienne,wart_pocz,zakres_parametr)$
(%i14) sol_faza:makelist([second(sol[i]),third(sol[i])],i,1,
    length(sol))$
(%i15) sol_x:makelist([first(sol[i]),second(sol[i])],i,1,
    length(sol))$
(%i16) sol_xdot:makelist([first(sol[i]),third(sol[i])],i,1,
    length(sol))$
(%i17) plot2d([discrete,sol_faza])$
(%i18) plot2d([[discrete,sol_x],[discrete,sol_xdot]],
    [legend,"x(t)","x'(t)"])$
```

W skrypcie przyjęto, nieco arbitralnie, że $\theta = 0.1$, $\chi = -0.25$, $\eta = 1$, $\kappa = 1$, $\omega = 1$. Na rysunku 11.7 pokazano portret fazowy oraz przebiegi czasowe $x(t)$ i $\dot{x}(t)$.



(a)



(b)

Rys. 11.7. Portret fazowy równania Duffinga (a) oraz przebiegi czasowe $x(t)$ i $\dot{x}(t)$ (b)

Ciekawym sposobem pokazania chaotycznego charakteru rozwiązań równania Duffinga są przekroje Poincaré⁵. Idea polega na wykonaniu przecięć przestrzeni fazowej $f(x, \dot{x}, t)$ w równych odstępach czasu T i rzutowaniu

⁵ Henry Poincaré (1854–1912) – matematyk, fizyk, astronom, przyrodnik, filozof.

punktów na płaszczyznę fazową (x, \dot{x}) . W przypadku układów chaotycznych uzyskany diagram jest bardzo złożony i ma strukturę fraktalną (rys. 11.8). Jeśli do równania (11.17) wprowadzić człon oznaczający tłumienie $\delta\dot{x}$, przyjmie ono postać zbliżoną do równania (11.18):

$$\ddot{x} + \delta\dot{x} + \eta x - \theta x^2 - \chi x^3 = \kappa \sin(\omega t). \quad (11.21)$$

W skrypcie wyznaczającym przekrój Poincaré przyjęto, że $\delta = 0.1$, do kreślenia atraktora wykorzystano polecenie `wxdraw2d()`. Zmiany parametrów w obu połączonych ze sobą skryptach pozwalają na symulację zachowania się równania (11.21) dla różnych wartości parametrów.


```
(%i19) %delta:0.1$
(%i21) iteracje:4000$ podzial:31$
(%i22) zakres_parametr:[t,0,iteracje*pi,%pi/podzial]$
(%i23) rownanie1:[x2,-%delta*x2-%eta*x1+%theta*x1^2+%chi*x1^3+
  %kappa*sin(%omega*t)]$
(%i24) atraktor:rk(rownanie1,zmienne,wart_pocz,zakres_parametr)$
(%i25) atrak10:create_list(atraktor[i],i,
  makelist(i*podzial,i,1,iteracje))$
(%i26) punktyatrak10:map(lambda([x],rest(x)),
  makelist(atrak10[i],i,1,iteracje))$
(%i27) draw2d(point_size=0.3,point_type=circle,
  xrange=[-4,4],yrange=[-7,4],xtics=1,ytics=1,color=blue,
  points(punktyatrak10),grid=true)$
```

11.3.2. Podejście analityczne – wspomagane Maximą

Symulacje komputerowe, możliwe do przeprowadzenia z wykorzystaniem skryptu pokazanego w poprzednim punkcie, pozostawiają jednak pewien niedosyt – uzyskane rozwiązania są (tylko?!) numeryczne. W literaturze [14] można odnaleźć rozwiązania analityczne równania:

$$\ddot{x} + \dot{x}\varepsilon k + x + \varepsilon\alpha x^3 = \varepsilon F \cos(\Omega t), \quad (11.22)$$

Niezależnie od Einsteina stworzył matematyczne podstawy szczególnej teorii względności. Twórca wielu znaczących prac z matematyki, filozofii, fizyki.



Rys. 11.8. Przekrój Poincaré dla równania (11.17)

gdzie: $\ddot{x} + x$ – człon odpowiadający za drgania nietłumione,
 $\varepsilon k \dot{x}$ – małe tłumienie,
 $\varepsilon \alpha x^3$ – mała nieliniowość,
 $\varepsilon F \cos(\Omega t)$ – małe wymuszenie harmoniczne,
 $\Omega = 1 + \varepsilon \omega$ – częstotliwość wymuszenia,
 ω – parametr dostrojenia.

Pojawiające się w równaniu (11.22) parametry k , α , F , ω to odpowiednio: tłumienie, nieliniowość, amplituda wymuszenia, parametr dostrojenia. Zakłada się, że $k, F > 0$ oraz α, ω należą do zbioru liczb rzeczywistych. Przyjęto również, że ε jest małą stałą większą od zera. Zakładamy, że rozwiązanie równania (11.22) ma postać:

$$x = A \cos(\theta) + \varepsilon B \cos(3\theta). \quad (11.23)$$

Podstawiając $\theta = \Omega t - \varphi$ i różniczkując dwukrotnie, a następnie podstawiając $\Omega t - \varphi = \theta$, otrzymujemy:

$$\dot{x} = -\Omega A \sin \theta - 3\Omega \varepsilon B \sin 3\theta, \quad (11.24)$$

$$\ddot{x} = \Omega^2 A \cos \theta - 9\Omega^2 \varepsilon B \cos 3\theta. \quad (11.25)$$

Odpowiedni fragment kodu w Maximie ma postać:

```
(%i1) kill(all)$ restet()$
(%i3) x:A*cos(%theta)+%epsilon*B*cos(3*%theta);
(%i4) xdot1:diff(subst(%Omega*t-%phi,%theta,x),t);
(%i5) xdot:subst(%theta,%Omega*t-%phi,xdot1);
```

```
(%i6) xddot1:diff(xdot1,t);
(%i7) xddot:subst(%theta,%Omega*t-%phi,xddot1);
```

W wyrażeniu odpowiadającym za siłę wymuszającą $P = \varepsilon F \cos(\Omega t)$ podstawiamy $\Omega t = \theta + \varphi$. Po przekształceniach otrzymujemy: $P = \varepsilon F(\cos(\phi) \cos(\theta) - \sin(\phi) \sin(\theta))$.

```
(%i8) P:%epsilon*F*cos(%Omega*t)$
(%i9) P:subst(%theta+%phi,%Omega*t,P)$
(%i10) P:trigexpand(P);
(%o10)
```

$$\varepsilon F(\cos(\phi) \cos(\theta) - \sin(\phi) \sin(\theta)) \quad (11.26)$$

W przyjętym sposobie rozwiązywania zakłada się, że wyrażenie $\ddot{x} + x$ odpowiadające za drgania nietłumione ma postać:

```
(%i11) eq1a:ratsimp(xddot+x);
(%o11)
```

$$(\varepsilon - 9\varepsilon\Omega^2) \cos(3\theta)B + (1 - \Omega^2) \cos(\theta)A. \quad (11.27)$$

Podstawiając w uzyskanym wyrażeniu $\Omega^2 = 1 + 2\varepsilon\omega + \varepsilon^2\omega^2$, otrzymujemy wyrażenie $\ddot{x} + x$ w postaci:

```
(%i12) eq1a:subst(1+2*%omega*%epsilon+%epsilon^2*%omega^2,
%Omega,eq1a);
(%o12)
```

$$\begin{aligned} & (-8\varepsilon - 36\varepsilon^2\omega - 54\varepsilon^3\omega^2 - 36\varepsilon^4\omega^3 - 9\varepsilon^5\omega^4) \cos(3\theta) B \\ & + (-4\varepsilon\omega - 6\varepsilon^2\omega^2 - 4\varepsilon^3\omega^3 - \varepsilon^4\omega^4) \cos(\theta) A. \end{aligned} \quad (11.28)$$

Zaniedbując dalej wielkości małe typu ε^2 i mniejsze, otrzymujemy ostatecznie formułę określającą wyrażenie $\ddot{x} + x$:

```
(%i13) eq1:(-8*%epsilon)*cos(3*%theta)*B
+(-4*%epsilon*%omega)*cos(%theta)*A
(%o13)
```

$$-8\varepsilon \cos(3\theta)B - 4\varepsilon\omega \cos(\theta)A. \quad (11.29)$$

W podobny sposób uprościmy wyrażenia $\varepsilon k\dot{x}$ oraz $\varepsilon\alpha x^3$, usuwając harmoniczne wyższych rzędów. W pierwszym wyrażeniu podstawiamy $\Omega = 1 + \varepsilon\omega$ i pomijamy $\varepsilon^2\omega$. W drugim wyrażeniu zaniedbujemy $\varepsilon \cos(3\theta)B$ jako wielkość bliską zeru oraz podstawiamy $(\cos(\theta))^3 = (1/4)\cos(3\theta) + (3/4)\cos(\theta)$.

(%i14) eq2: -%epsilon*sin(%theta)*k*A;

Dla $\varepsilon k \dot{x}$ mamy:

$$-k\varepsilon \sin(\theta)A,$$

(%i15) eq31: %epsilon*%alpha*((A*cos(%theta))^2+(A*cos(%theta))^3);

eq32: subst((1/4)*cos(3*%theta)+(3/4)*cos(%theta),

cos(%theta)^3, eq31);

eq3: subst((1/2)*cos(2*%theta)+1/2, cos(%theta)^2, eq32);

zaś dla $\varepsilon \alpha x^3$ mamy:

$$\alpha\varepsilon\left(\left(\frac{3\cos(\theta)}{4} + \frac{\cos(3\theta)}{4}\right)A^3 + \left(\frac{1}{2} + \frac{\cos(2\theta)}{2}\right)A^2\right).$$

W tym miejscu możemy przystąpić do budowy równania Duffinga (11.22), podstawiając uzyskane wyrażenia: $\ddot{x} + x \rightarrow eq1$, $\varepsilon k \dot{x} \rightarrow eq2$, $\varepsilon \alpha x^3 \rightarrow eq3$ i $\varepsilon F \cos(\Omega t) \rightarrow P$.

(%i16) Duffing: eq1+eq2+eq3=P;

(%o17)

$$8\varepsilon \cos(3\theta)B + \alpha\varepsilon \left(\left(\frac{\cos(2\theta)}{2} + \frac{1}{2} \right) A^2 + \left(\frac{\cos(3\theta)}{4} + \frac{3\cos(\theta)}{4} \right) A^3 \right) - \varepsilon \sin(\theta)kA - 4\varepsilon\omega \cos(\theta)A = \varepsilon(\cos(\phi) \cos(\theta) - \sin(\phi) \sin(\theta))F. \quad (11.30)$$

Pora na najważniejszą część omawianej metody:

- dzielimy równanie (11.30) stronami przez ε ,
- porównujemy odpowiednie współczynniki, przy $\sin(\theta)$, $\cos(\theta)$ i $\cos(3\theta)$ po obu stronach równania (11.30)
 1. $\cos(\theta) \rightarrow -4\omega A + \frac{3}{4}\alpha A^3 = F \cos(\phi)$,
 2. $\sin(\theta) \rightarrow kA = F \sin(\phi)$,
 3. $\cos(3\theta) \rightarrow -8B + \alpha A^3/4 = 0$ - zanedbujemy.

Kolejne polecenia przybliżają nas do zdefiniowania zależności pomiędzy amplitudą A a wartościami parametru dostrojenia ω .

(%i18) eqD1: F*sin(%phi)=k*A;

(%i19) eqD2: F*cos(%phi)=A*((3/4)*%alpha*A^2-4*%omega);

Podnosimy oba równania stronami do kwadratu, dodajemy do siebie oraz upraszczamy trygonometrycznie:

(%i20) eqD: trigsimp(eqD1^2+eqD2^2);

(%o20)

$$F^2 = \frac{(16k^2 + 256\omega^2)A^2 - 96\alpha\omega A^4 + 9\alpha^2 A^6}{16}. \quad (11.31)$$


Uzyskana zależność jest uwikłana, możemy jednak dokonać graficznej analizy wpływu parametru dostrojenia na amplitudę ruchu drgającego. Załóżmy na początek, że współczynnik $\alpha = 0$, co oznacza, że równanie (11.22) opisuje prosty ruch drgający, dodatkowo ustalamy wartości $k = 1$ i $F = 1$. Dalej kreślimy charakterystyki amplitudowo-częstotliwościowe, podobnie postępujemy dla innych wartości parametru α .

```
(%i21) load(implicit_plot)$
(%i22) k:1$ F:1$ %alpha:0$
(%i23) wximplicit_plot (eqD, [%omega, -4, 4], [A, 0, 1.5],
                        [gnuplot_preamble, "set zeroaxis"]);
(%i24) k:1$ F:1$ %alpha:-5$
(%i25) wximplicit_plot (eqD, [%omega, -4, 4], [A, 0, 1.5],
                        [gnuplot_preamble, "set zeroaxis"]);
(%i26) k:1$ F:1$ %alpha:5$
(%i27) wximplicit_plot (eqD, [%omega, -4, 4], [A, 0, 1.5],
                        [gnuplot_preamble, "set zeroaxis"]);
```

Na rysunkach 11.9 a, b, c pokazano uzyskane charakterystyki amplitudowo-częstotliwościowe. Łatwo dostrzec ciekawą, wręcz zaskakującą właściwość analizowanego równania (11.22). W przypadku kiedy $\alpha = 0$, mamy do czynienia z równaniem opisującym ruch wymuszony tłumiony (11.12). Zmieniając wartość parametru dostrojenia ω , opisującego zgodność częstości wymuszenia z częstością własną układu drgającego, można dostrzec, że dla pewnej jego wartości pojawiają się drgania rezonansowe.

Zachowanie układu opisanego równaniem (11.22) dla niezerowych wartości α jest już o wiele bardziej zaskakujące. Jeśli przyjąć przykładowo $\alpha = -5$, zmiana parametru dostrojenia ω polegająca na jego wzroście od -4 do 4 pokazuje, że w pobliżu wartości $\omega = -1$ następuje skokowy wzrost amplitudy drgań do wartości odpowiadającej rezonansowi. Dalsze zwiększanie wartości parametru ω powoduje spadek wartości amplitudy ruchu drgającego. Podobne zjawisko możemy dostrzec dla $\alpha = 5$. Zwiększanie wartości parametru dostrojenia ω od -4 do 4 powoduje dla $\omega = 1$ osiągnięcie rezonansu. Jednak już niewielkie przekroczenie tej wartości powoduje skokowy spadek amplitudy drgań.

Pokazane na rys. 11.9 charakterystyki amplitudowo-częstotliwościowe są „przekrojami” wyrażenia (11.31) dla ustalonych wartości parametru α . Wykreślenie zależności amplitudy A w funkcji α i ω wymaga usunięcia z pamięci (%i28) ustalonych wartości, przypisanych zmiennym α i ω . Następn-



Rys. 11.9. Charakterystyki amplitudowo-częstotliwościowe dla różnych wartości α , $\alpha = 0$ (a), $\alpha = -5$ (b), $\alpha = 5$ (c) oraz wykres $f(A, \omega, \alpha)$ (d)

nie, po wczytaniu pakietu `draw`, definiujemy funkcję uwikłaną poleceniem `implicit()` oraz wykreślamy ją poleceniem `draw3d()` (rys. 11.9d).

```
(%i28) kill(%alpha,%omega); load(draw)$
(%i29) k:1;F:1;
(%i30) AC:implicit(ev(eqD),%omega,-2,2,%alpha,-5,5,A,0,1);
(%i31) wxdraw3d(file_name="c:/ścieżka do ../nazwa rysunku",
                terminal='eps_color,
                x_voxel = 30,
                y_voxel = 20,
                z_voxel = 20,
```

```
enhanced3d = true,  
xlabel="omega",  
ylabel="alfa",  
zlabel="A",  
surface_hide = true,  
user_preamble="set hidden3d",  
color=blue,AC)$
```